

# Sage Tutorium 1

## Inhaltsverzeichnis

1. [Sage als Taschenrechner](#)
2. [Symbolisches Rechnen](#)
3. [Plotten von Funktionen](#)
4. [Interaktive Programme](#)
5. [Kontrollstrukturen](#)
6. [Definieren von Funktionen](#)

## Sage als Taschenrechner

Man kann Sage einfach als **sehr** mächtigen Taschenrechner verwenden. Es werden dabei Zahlen mit beliebiger Genauigkeit unterstützt, ebenso Brüche und komplexe Zahlen. Für viele Berechnungen werden exakte Ergebnisse geliefert.

Um eine Rechnung auszuführen, geben wir die Rechnung in eine Zelle ein und drücken dann **Umschalt + Enter** um die Berechnung zu starten.

```
1 + 1
```

2

```
(2 * 2 + 5)/3
```

3

## Exponentiation

```
2^10
```

1024

## Langzahlarithmetik

```
2^1000
```

1071508607186267320948425049060001810561404811705533607443750388376  
5105112493612249319837881569585812759467291755314682518714528569231  
0435984577574698574803934567774824230985421074605062371141877954182  
5304647498358194126739876755916554394607706291457119647768654216766  
429831652624386837205668069376

## Bruchrechnen

```
1/6 + 7/12
```

3/4

## Komplexe Zahlen

```
sqrt(-4)
```

2\*I

```
I^2
```

-1

Realteil

```
real((1 + I)/(1 - I))
```

0

Imaginärteil

```
imag((1 + I)/(1 - I))
```

1

## Exakte Auswertung mathematischer Funktionen

```
sin(pi/2)
```

1

```
sin(pi/3)
```

$\sqrt{3}/2$

```
sin(pi/4)
```

$1/\sqrt{2}$

Um Ungleichungen exakt auszuwerten, müssen wir sie in einen Wahrheitswert umwandeln.

```
bool(sqrt(2) < 2 * sin(pi/3))
```

True

## Gleitkommaarithmetik mit beliebiger Genauigkeit

```
numerical_approx(pi, digits = 400)
```

```
3.14159265358979323846264338327950288419716939937510582097494459230
8164062862089986280348253421170679821480865132823066470938446095505
2231725359408128481117450284102701938521105559644622948954930381964
2881097566593344612847564823378678316527120190914564856692346034861
4543266482133936072602491412737245870066063155881748815209209628292
4091715364367892590360011330530548820466521384146951941511609
```

Dasselbe in objektorientierter Notation

```
(pi + e^2).n(digits = 1000)
```

```
10.5306487525204434656930708438545106973774849699269531450620724148
0390202365266762012347310421238862755901247778761172770978448302331
5560620375162152655889235040699830934927480795097557049963701696378
6984588438743376671107308831095292457066858818372749804549942236496
9792226388943380567355552448789045405830291010381204795079692408512
7801498812897048093035647254337983403014542652297161592863442315991
3351013859712482744139603303671979283928124150602143815127097154269
7724018260646790180240831766089645303536522834565892899274792918001
0093729617333369278105300058305767541627736043428953362070924906013
3387951819997375030900275864385188370292633170728725818129003756751
9354611897429482940373819664763755114126142981090362351601313246056
8645832842067385414023361513679088972798600438648896389142901008412
9096137351724467366899876816250727486428981207957815392351866099129
8539990136393143290671564305390008578501753092962932683468669173048
5948408910465623063448527807274335625240795500791468915150064000945
0027
```

Bei längeren Berechnungen ist es sinnvoll sich zuerst einen



```
expr1 = (x + y + z)^3
expr1
```

```
(z + y + x)^3
```

```
type(expr1); parent(expr1)
```

```
<class 'sage.calculus.calculus.SymbolicArithmetic'>
Symbolic Ring
```

## Anzeige in mathematischer Notation

```
view(expr1)
```

```
(z + y + x)3
```

## Ausmultiplizieren

```
view(expand(expr1))
```

```
 $z^3 + 3yz^2 + 3xz^2 + 3y^2z + 6xyz + 3x^2z + y^3 + 3xy^2 + 3x^2y + x^3$ 
```

## Faktorisieren

```
expr2 = (x^2 - 2 * x * y + y^2)
view(expr2)
```

```
 $y^2 - 2xy + x^2$ 
```

```
view(factor(expr2))
```

```
 $(x - y)^2$ 
```

## Substitution von Variablen und Termen

```
expr3 = expr2(y = sin(z^2))
view(expr3)
```

```
 $(\sin(z^2))^2 - 2x \sin(z^2) + x^2$ 
```

```
expr3.subs_expr(sin(z^2) == y)
```

```
 $y^2 - 2*x*y + x^2$ 
```

## Differentialrechnung

```
diff(x * sin(x), x)
```

```
 $\sin(x) + x*\cos(x)$ 
```

## Polynome

Wir konstruieren einen Datentyp für Polynome in der Variablen  $t$ , und Koeffizienten aus den Rationalen Zahlen ( $\mathbb{Q}$ ).

```
P.<t> = QQ[]
type(P)
```

```
<class
'sage.rings.polynomial.polynomial_ring.PolynomialRing_field'>
```

gen() gibt die Variable des Polynoms zurück.

```
P.gen()
```

```
t
```

Wir definieren uns einige Polynome

```
p1 = (1/2 * t - 3/2) * (t - 1/4)
view(p1)
```

```
 $\frac{1}{2}t^2 - \frac{13}{8}t + \frac{3}{8}$ 
```

roots() berechnet die Nullstellen eines Polynoms zusammen mit ihren Vielfachheiten.

```
p1.roots()
```

```
[(3, 1), (1/4, 1)]
```

Probe durch Einsetzen.

```
p1(3); p1(1/4)
```

```
0  
0
```

degree() berechnet den Grad des Polynoms.

```
p1.degree()
```

```
2
```

```
p2 = (1 + t - t^2)^3  
view(p2)
```

```
-t^6 + 3t^5 - 5t^3 + 3t + 1
```

```
parent(p2)
```

```
Univariate Polynomial Ring in t over Rational Field
```

Standardmäßig berechnet roots() nur Nullstellen in der selben Grundmenge in der die Koeffizienten des Polynoms liegen. In unserem Fall  $\mathbb{Q}$ . Unser Polynom hat keine rationalen Nullstellen.

```
p2.roots()
```

```
[]
```

Wir können die Grundmenge aber auch explizit angeben. Hier berechnen wir alle reellen Nullstellen.

```
p2.roots(ring = RR)
```

```
[(-0.618031057174341, 1), (1.61804465415111, 1)]
```

Um sämtliche Nullstellen zu bekommen, brauchen wir komplexe Zahlen.

```
p2.roots(ring = CC)
```


```
[(-0.618031057174341, 1), (1.61804465415111, 1), (1.61802865604929  
9.23641138258941e-6*I, 1), (1.61802865604929 -  
9.23641138258941e-6*I, 1), (-0.618035454537672 +  
2.53887260392275e-6*I, 1), (-0.618035454537672 -  
2.53887260392275e-6*I, 1)]
```

## Plotten von Funktionen

Sage verfügt über vielfältige High-level Funktionen zum Plotten von Funktionen.


Wir plotten das Polynom  $p_2 = -t^6 + 3t^5 - 5t^3 + 3t + 1$  im Intervall  $[-1, 2]$ .

```
plot(p2, -1, 2)
```




Plot einer Funktion mit Polstellen. Es sind leider keine Details sichtbar.

```
plot((sin(5 * x) - 1)/(2 * x^2 - 3 * x - 2), (x, -3, 3), rgbcolor = 'red')
```



Für ein besseres Ergebnis müssen wir den den Ausschnitt explizit auswählen.

```
plot2 = plot((sin(5 * x) - 1)/(2 * x^2 - 3 * x - 2), (x, -3, 3),
rgbcolor = 'red')
plot2.show(ymin = -5, ymax = 5)
```



## Interaktive Programme

Sage bietet die Möglichkeit kleine interaktive Programme in das Notebook einzubinden (Siehe die Dokumentation zu `interact`). Weitere Beispiele für `interact` finden im [Sage-Wiki](#).

Das folgende Beispiel visualisiert die Taylorpolynome der Funktion  $f(x) = \sin(x) \cdot e^{-x}$ .

```

var('x')
f = sin(x) * e^(-x)
p = plot(f, -1, 5, thickness = 2)
html('<h2>Taylor Polynome</h2>')

@interact
def _(x0 = input_box(0, label = "x0="),
    order = slider(0, 12, 1, label = "Ordnung: ", default = 3)):

    ft = f.taylor(x, x0, order)
    pt = plot(ft, -1, 5, color = 'green', thickness = 2)
    dot = point((x0, f(x0)), pointsize = 80, rgbcolor = (1, 0, 0))
    html('$f(x) = %s$' % latex(f))
    html('$\hat{f}(x;%s) = %s + \mathcal{O}(x^{%s})$' % (latex(x0),
    latex(ft), order + 1))
    (dot + p + pt).show(ymin = -.5, ymax = 1)

```


## Taylor Polynome

x0= 0

Ordnung:

$$f(x) = e^{-x} \sin(x)$$

$$\hat{f}(x; 0) = x - x^2 + \frac{x^3}{3} + \mathcal{O}(x^4)$$





# Datentypen

Die wichtigsten Datentypen in Sage/Python sind:

- [Tupel](#)
- [Listen](#)
  - [Erzeugen von Listen](#)
  - [Kurznotation für Arithmetische Progressionen](#)
  - [List Comprehensions](#)
- [Strings](#)
- [Dictionaries](#)
- [Mengen](#)

## Tupel

Tupel sind ähnlich wie Listen, nur können einmal erstellte Tupel nicht mehr verändert werden (immutable). Tupel sind besonders nützlich, um mehrere Werte gleichzeitig von einer Funktion zurückzugeben.

```
a = 1, 2, 3
a; type(a)
(1, 2, 3)
<type 'tuple'>
```

```
a[0]; a[1]; a[2]
1
2
3
```

Die Funktion `xgcd(a, b)` berechnet den kleinsten gemeinsamen Teiler von `a` und `b` mit Hilfe des erweiterten Euclidschen Algorithmus und gibt das Ergebnis als Tupel von drei Zahlen zurück.

```
t = xgcd(14, 25)
t
(1, -16, 9)
```

Man kann die Elemente des Tupels auch direkt einzelnen Variablen zuweisen (Tupel unpacking).

```
a1, a2, a3 = xgcd(14, 25)
a2
-16
```

Simultanes vertauschen zweier Variablen

```
t1 = 10
t2 = cos(x)
t1; t2
10
cos(x)
```

```
t1, t2 = t2, t1
t1; t2
cos(x)
10
```

## Listen

Listen sind wahrscheinlich der wichtigste Datentyp in Sage/Python. Der wichtigste Unterschied zu Tupeln ist dass Listen veränderbar (mutable) sind. Sie können Elemente zu Listen hinzufügen, löschen, verändern. Das ist bei Tupeln nicht möglich.

### Einige Operationen auf Listen:

- `[]` erzeugt die leere Liste
- `len(L)` bestimmt die Länge der Liste
- `L[k]` greift auf das Element an der Stelle `k` zu. Allerdings beginnt die Nummerierung bei 0. Die gültigen Indizes für eine Liste der Länge `n` sind also die Werte `0,...n-1`.
- `L[-k]` greift auf das `k`-te Element der Liste gezählt von hinten zu. `-n,...,-1`, (das entspricht `(n)-n,...(n)-1`) sind hier die gültigen Werte.
- `L[i:k]` greift auf den Bereich `L[i]...L[k-1]` zu.
- `L.append(elem)` fügt `elem` ans Ende der Liste `L` hinzu.
- `L1+L2`, die Listen `L1` und `L2` werden aneinandergelängt.

```
L = [1, 2, 3, 4, 5, 6, 7, 8, 9, 10]
```

```
L
```

```
[1, 2, 3, 4, 5, 6, 7, 8, 9, 10]
```

```
len(L)
```

```
10
```

```
print L[0] # erstes Element
print L[1] # zweites Element
print L[-1] # letztes Element
print L[2:5] # ein Teil der Liste
```

```
1
```

```
2
```

```
10
```

```
[3, 4, 5]
```

### Hinzufügen von Elementen

```
L.append(11)
print "Liste:", L
print "Anzahl der Elemente:", len(L)
```

```
Liste: [1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11]
```

```
Anzahl der Elemente: 11
```

### Ersetzen von Elementen und Teillisten

```
L[0] = "Hallo"
```

```
L
```

```
['Hallo', 2, 3, 4, 5, 6, 7, 8, 9, 10, 11]
```

```
L[1:3] = 1/2
```

```
L
```

```
['Hallo', 1/2, 4, 5, 6, 7, 8, 9, 10, 11]
```

```
L[2] = [pi^2/6, e]
```

```
L
```

```
['Hallo', 1/2, [pi^2/6, e], 5, 6, 7, 8, 9, 10, 11]
```

Sehr nützlich ist die Funktion `zip`, die ein Tupel von Listen in eine Liste von Tupeln verwandelt.

```
zip([1, 2, 3], ['a', 'b', 'c'], [sin, cos, tan])
```



```
[x^2 for x in range(10)]  
[0, 1, 4, 9, 16, 25, 36, 49, 64, 81]
```

```
var('x')  
[sin(x).diff(i) for i in [0..4]]  
[sin(x), cos(x), -sin(x), -cos(x), sin(x)]
```

```
[x^2 for x in [1..20] if x % 2 == 0]  
[4, 16, 36, 64, 100, 144, 196, 256, 324, 400]
```

```
[(x,y) for x in [-3..3] for y in [-2..2] if -1 <= x-y <= 1]  
[(-3, -2), (-2, -2), (-2, -1), (-1, -2), (-1, -1), (-1, 0), (0, -1),  
(0, 0), (0, 1), (1, 0), (1, 1), (1, 2), (2, 1), (2, 2), (3, 2)]
```

## Strings

Strings sind im wesentlichen Listen von Buchstaben, allerdings sind Strings aus Performancegründen **nicht mutable**, genauso wie Tupel.

```
s1 = "computer"  
s2 = "mathematik"  
s3 = "informatik"
```

```
s4 = s1.capitalize() + s2  
s4  
'Computermathematik'
```

```
s = s4 + " (%s)" % s3.capitalize()  
s  
'Computermathematik (Informatik)'
```

```
len(s)  
31
```

```
s[8:18].capitalize()  
'Mathematik'
```

```
s.split()  
['Computermathematik', '(Informatik)']
```

```
s.upper()  
'COMPUTERMATHEMATIK (INFORMATIK)'
```

```
suchstring = "Informatik"  
istart=s.find(suchstring)  
istart  
20
```

```
s[istart:istart + len(suchstring)]  
'Informatik'
```

## Dictionaries

Ein Dictionary speichert Zuordnungen von je einem Wert zu einem Schlüsselwert

Einige nützliche Funktionen:

- {}, erzeugt ein leeres Dictionary
- d[s], gibt das Element mit Schlüssel s zurück
- keys(), gibt die Liste der Schlüssel zurück
- values(), die Liste der Werte
- items(), erzeugt eine Liste von allen (Schlüssel, Wert) Paaren
- has\_key(s), gibt zurück, ob der Schlüssel s vorhanden ist

```
huss = {'name': 'Wilfried Huss', 'office': 'C305', 'email':
'huss@finanz.math.tugraz.at'}
huss
```

```
{'name': 'Wilfried Huss', 'office': 'C305', 'email':
'huss@finanz.math.tugraz.at'}
```

```
huss['name']
'Wilfried Huss'
```

```
huss.keys()
['name', 'office', 'email']
```

```
huss.values()
['Wilfried Huss', 'C305', 'huss@finanz.math.tugraz.at']
```

```
huss.items()
[('name', 'Wilfried Huss'), ('office', 'C305'), ('email',
'huss@finanz.math.tugraz.at')]
```

```
huss.has_key('name'), huss.has_key('nachname')
(True, False)
```

```
for (key, value) in huss.iteritems():
    print key, ":", value
    name : Wilfried Huss
    office : C305
    email : huss@finanz.math.tugraz.at
```

```
primzahlen = {}
p = 1;
for i in [1..30]:
    p = next_prime(p)
    primzahlen[p] = i
```

```
primzahlen
{2: 1, 3: 2, 5: 3, 7: 4, 11: 5, 13: 6, 17: 7, 19: 8, 23: 9, 29: 10,
31: 11, 37: 12, 41: 13, 43: 14, 47: 15, 53: 16, 59: 17, 61: 18, 67:
19, 71: 20, 73: 21, 79: 22, 83: 23, 89: 24, 97: 25, 101: 26, 103:
27, 107: 28, 109: 29, 113: 30}
```

```
p = 109
print "%d ist die %d. Primzahl" % (p, primzahlen[p])
109 ist die 29. Primzahl
```

## Mengen

```
s1 = set([1..10])
s2 = set([5..15])
s3 = set([1, 1, 1, 1])
s1; s2; s3
```

```
set([1, 2, 3, 4, 5, 6, 7, 8, 9, 10])  
set([5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15])  
set([1])
```

```
s1.union(s2)
    set([1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15])
s1.intersection(s2)
    set([5, 6, 7, 8, 9, 10])
s1.symmetric_difference(s2)
    set([1, 2, 3, 4, 11, 12, 13, 14, 15])
```

## Kontrollstrukturen

[Python](#) die Programmiersprache von Sage verwendet Einrückungen zur Definition von Blöcken.

### If-Statement

```
a = -4
if a > 0:
    print a, "ist positiv"
elif a < 0:
    print a, "ist negativ"
else:
    print a, "ist null"
-4 ist negativ
```

### For-Schleife

```
for i in [1,2,3,4]:
    j = i^2
    print j
1
4
9
16
```

### While-Schleife

```
i = 1
while i < 20:
    if i.is_prime():
        print "%2d ist eine Primzahl" % i
    i += 1
2 ist eine Primzahl
3 ist eine Primzahl
5 ist eine Primzahl
7 ist eine Primzahl
11 ist eine Primzahl
13 ist eine Primzahl
17 ist eine Primzahl
19 ist eine Primzahl
```

## Definieren von Funktionen

```
def absolutbetrag(x):  
    if x < 0:  
        return -x  
    else:  
        return x
```

```
absolutbetrag(-3); absolutbetrag(6)
```

3  
6

Selbstdefinierte Funktionen funktionieren automatisch für alle Typen, die alle in der Funktion verwendeten Methoden und Operationen unterstützen.

In unserem Beispiel:

- Vergleichsoperator <
- Negation -

```
R = RealField(prec = 200)  
absolutbetrag(R.random_element(-10, -5))
```

6.3143950641973070900223374505923141912547366595474086604627